

INVESTIGACIÓN DE OPERACIONES
PRÁCTICA DE LABORATORIO #5
Modelado matemático

Nombre: CARLOS EDUARDO SANCHEZ TORRES Fecha: 25/08/2020

1. En un Estado se tienen cinco poblaciones en que asisten niños y niñas a tres distintas escuelas primarias. El número de niños y niñas en cada población se presenta en la tabla siguiente:

POBLACIÓN	NIÑOS	NIÑAS
1	35	15
2	25	10
3	90	5
4	10	115
5	15	30
TOTAL	175	175

Las capacidades de las escuelas son:

ESCUELA	A	B	C	TOTAL
CAPACIDAD	170	80	100	350

Las distancias, en kilómetros, entre las distintas poblaciones y las escuelas son:

POBLACIÓN	E S C U E L A		
	A	B	C
1	3	7	9
2	4	4	12
3	6	4	10
4	7	5	4
5	4	7	3

Se desea asignar niños y niñas a las escuelas de manera tal que el número total de kilómetros viajados sea mínimo y que se cumplan las siguientes condiciones: (a) Todos los niños y las niñas van a la escuela, (b) Las escuelas no rebasan su capacidad, (c) Cada escuela tiene entre 40% y 60% de niños. Formule el modelo matemático de Programación Lineal para este problema y encuentre la asignación óptima desarrollando un programa en LINGO (con datos estructurados).

a. Modelo matemático.

x_{ijk} : número de i en población j que van a escuela k

$i = 1, 2$ (niño o niña); $j = 1..5$; $k = A, B, C$

$$\text{Min } Z = 3(x_{111} + x_{211}) + 7(x_{112} + x_{212}) + 9(x_{113} + x_{213}) + \dots + 3(x_{153} + x_{253})$$

Todos los niños y niñas van a la escuela:

$$\sum_k x_{11k} = 35 \text{ y } \sum_k x_{21k} = 15$$

...

$$\sum_k x_{1jk} = \text{niños en población } j \text{ y } \sum_k x_{2jk} = \text{niñas en población } j$$

$$\sum_j x_{1jA} + x_{1jB} + x_{1jC} = 175$$

$$\sum_j x_{2jA} + x_{2jB} + x_{2jC} = 175$$

Las escuelas no rebasan su capacidad:

$$\sum_j x_{1jA} + x_{2jA} \leq 170 (\text{escuela A})$$

$$\sum_j x_{1jB} + x_{2jB} \leq 80 (\text{escuela B})$$

$$\sum_j x_{1jC} + x_{2jC} \leq 100 (\text{escuela C})$$

Cada escuela tiene entre 40% y 60% de niños:

$$170 * 0.4 \leq \sum_j x_{1jA} \leq 170 * 0.6$$

$$80 * 0.4 \leq \sum_j x_{1jB} \leq 80 * 0.6$$

$$100 * 0.4 \leq \sum_j x_{1jB} \leq 100 * 0.6$$

b. Programa en LINGO

```
SETS:
    POBLACION: NINAS, NINOS;
    ESCUELA: CAPACIDAD;
    TIPOS;
    FORMULA(TIPOS, POBLACION, ESCUELA): KM, X;
ENDSETS

DATA:
    ESCUELA = A B C;
    CAPACIDAD = 170 80 100;
    POBLACION = 1 2 3 4 5;
    TIPOS =      NINOS NINAS;
    NINOS =      35 25 90 10 15;
    NINAS =      15 10 5 115 30;
    KM      =      3 7 9
                   4 4 12
                   6 4 10
                   7 5 4
                   4 7 3
                   3 7 9
                   4 4 12
                   6 4 10
                   7 5 4
                   4 7 3;
ENDDATA

MIN = @SUM(FORMULA: KM*X);

@FOR( POBLACION(i): @SUM(ESCUELA(j): X(1, i, j)) = NINOS(i));
@FOR( POBLACION(i): @SUM(ESCUELA(j): X(2, i, j)) = NINAS(i));
@FOR( TIPOS(i): @SUM(POBLACION(j): @SUM(ESCUELA(k): X(i, j, k)) ) = 175);
@FOR( ESCUELA(i) : @SUM(POBLACION(j): X(1,j,i)+X(2,j,i)) <= CAPACIDAD(i));
@FOR( ESCUELA(i) : @SUM(POBLACION(j): X(1,j,i)) <= CAPACIDAD(i)*0.6);
@FOR( ESCUELA(i) : @SUM(POBLACION(j): X(1,j,i)) >= CAPACIDAD(i)*0.4);
```

Objective value:

1590.000

X(NINOS, 1, A)	35.00000	0.000000
X(NINOS, 1, B)	0.000000	6.000000
X(NINOS, 1, C)	0.000000	2.000000
X(NINOS, 2, A)	25.00000	0.000000
X(NINOS, 2, B)	0.000000	2.000000
X(NINOS, 2, C)	0.000000	4.000000
X(NINOS, 3, A)	27.00000	0.000000
X(NINOS, 3, B)	48.00000	0.000000
X(NINOS, 3, C)	15.00000	0.000000
X(NINOS, 4, A)	0.000000	7.000000
X(NINOS, 4, B)	0.000000	7.000000
X(NINOS, 4, C)	10.00000	0.000000
X(NINOS, 5, A)	0.000000	5.000000
X(NINOS, 5, B)	0.000000	10.00000
X(NINOS, 5, C)	15.00000	0.000000
X(NINAS, 1, A)	15.00000	0.000000
X(NINAS, 1, B)	0.000000	6.000000
X(NINAS, 1, C)	0.000000	9.000000
X(NINAS, 2, A)	10.00000	0.000000
X(NINAS, 2, B)	0.000000	2.000000
X(NINAS, 2, C)	0.000000	11.00000
X(NINAS, 3, A)	0.000000	0.000000
X(NINAS, 3, B)	5.000000	0.000000
X(NINAS, 3, C)	0.000000	7.000000
X(NINAS, 4, A)	28.00000	0.000000
X(NINAS, 4, B)	27.00000	0.000000
X(NINAS, 4, C)	60.00000	0.000000
X(NINAS, 5, A)	30.00000	0.000000
X(NINAS, 5, B)	0.000000	5.000000
X(NINAS, 5, C)	0.000000	2.000000

2. Una empresa desea fabricar un artículo de plástico que es usado en la industria eléctrica y automovilística. Puede elaborarse en tres tamaños distintos: **chico, mediano y grande**. Estos artículos pueden ser elaborados en cualquiera de las tres plantas, **A, B y C**, que tiene la empresa. Las capacidades de fabricación diaria de estas plantas son independientes de la combinación de artículos que se produzcan. El principal recurso que limita la producción de estos artículos es el agua de enfriamiento para las máquinas.

La compañía desea determinar las **cantidades de cada uno de estos artículos en cada una de sus plantas por día** de manera de maximizar sus ganancias netas. Debe cumplirse además que, la fracción

$$p = \frac{\text{capacidad programada}}{\text{capacidad de la planta}}$$

sea igual para todas las plantas.

En la siguiente tabla se proporciona toda la información pertinente.

TIPO DE ARTÍCULO	PLANTAS			Agua por pieza, litros	Ventas diarias	Ganancia Unitaria, pesos
	A	B	C			
CHICO	X1	X2	X3	40	450	60
MEDIANO	X4	X5	X6	70	900	100
GRANDE	X7	X8	X9	90	700	120
Capacidad diaria	600	800	300			
Disponibilidad de agua, m3	40	30	18			

i) Construya un modelo matemático de Programación Lineal para este problema.

x_1, x_2, x_3 : cantidad diaria de artículos chicos fabricados en planta A, B, C
 x_4, x_5, x_6 : cantidad diaria de artículos medianos fabricados en planta A, B, C
 x_7, x_8, x_9 : cantidad diaria de artículos grandes fabricados en planta A, B, C

$$\text{Max } Z = 60 \sum_{i=1}^3 x_i + 100 \sum_{i=4}^6 x_i + 120 \sum_{i=7}^9 x_i$$

Ventas diarias:

$$x_1 + x_2 + x_3 \leq 450$$

$$x_4 + x_5 + x_6 \leq 900$$

$$x_7 + x_8 + x_9 \leq 700$$

Capacidad diaria:

$$x_1 + x_4 + x_7 \leq 600$$

$$x_2 + x_5 + x_8 \leq 800$$

$$x_3 + x_6 + x_9 \leq 300$$

Disponibilidad de agua:

$$40x_1 + 70x_4 + 90x_7 \leq 40000$$

$$40x_2 + 70x_5 + 90x_8 \leq 30000$$

$$40x_3 + 70x_6 + 90x_9 \leq 18000$$

Misma proporción:

$$(x_1 + x_4 + x_7)/600 =$$

$$(x_2 + x_5 + x_8)/800 =$$

$$(x_3 + x_6 + x_9)/300$$

ii) Realice un programa en Lingo no estructurado para obtener la solución óptima.

```
MAX = 60*X1+60*X2+60*X3+100*X4+100*X5+100*X6+120*X7+120*X8+120*X9;  
X1+X2+X3 <= 450;  
X4+X5+X6 <= 900;  
X7+X8+X9 <= 700;  
X1+X4+X7 <= 600;  
X2+X5+X8 <= 800;  
X3+X6+X9 <= 300;  
40*X1+70*X4+90*X7 <= 40000;  
40*X2+70*X5+90*X8 <= 30000;  
40*X3+70*X6+90*X9 <= 18000;  
X1+X4+X7 = (6/8)*(X2+X5+X8);  
X1+X4+X7 = (2)*(X3+X6+X9);
```

Objective value:

123116.1

Variable	Value	Reduced Cost
X1	0.000000	27.32143
X2	450.0000	0.000000
X3	0.000000	27.32143
X4	97.32143	0.000000
X5	171.4286	0.000000
X6	148.6607	0.000000
X7	368.7500	0.000000
X8	0.000000	18.21429
X9	84.37500	0.000000

iii) Realice un programa en Lingo estructurado para obtener la solución óptima.

```
SETS:
    PLANTA: CAPACIDAD, DISPONIBILIDAD_DE_AGUA;
    ARTICULO: AGUA_POR_PIEZA, VENTAS, GANANCIAS;
    PRODUCCION(PLANTA, ARTICULO): CANTIDAD;
ENDSETS

DATA:
    PLANTA = A B C;
    CAPACIDAD = 600 800 300;
    DISPONIBILIDAD_DE_AGUA = 40000 30000 18000;
    ARTICULO = CHICO MEDIANO GRANDE;
    AGUA_POR_PIEZA = 40 70 90;
    VENTAS = 450 900 700;
    GANANCIAS = 60 100 120;
ENDDATA

MAX = @SUM(PRODUCCION(I,J): GANANCIAS(J)*CANTIDAD(I,J));

@FOR(ARTICULO(I):
    @SUM(PLANTA(J): CANTIDAD(J,I)) <= VENTAS(I)
);

@FOR(PLANTA(I):
    @SUM(ARTICULO(J): CANTIDAD(I,J)) <= CAPACIDAD(I)
);

@FOR(PLANTA(I):
    @SUM(ARTICULO(J): AGUA_POR_PIEZA(J)*CANTIDAD(I,J)) <= DISPONIBILIDAD_DE_AGUA(I)
);

@FOR(PLANTA(I):
    (CAPACIDAD(1)/CAPACIDAD(I))*@SUM(ARTICULO(J): CANTIDAD(I,J)) = @SUM(ARTICULO(J):
    CANTIDAD(1,J))
);
```

Objective value: 123116.1

Variable	Value	Reduced Cost
GANANCIAS(CHICO)	60.00000	0.000000
GANANCIAS(MEDIANO)	100.0000	0.000000
GANANCIAS(GRANDE)	120.0000	0.000000
CANTIDAD(A, CHICO)	0.000000	27.32143
CANTIDAD(A, MEDIANO)	97.32143	0.000000
CANTIDAD(A, GRANDE)	368.7500	0.000000
CANTIDAD(B, CHICO)	450.0000	0.000000
CANTIDAD(B, MEDIANO)	171.4286	0.000000
CANTIDAD(B, GRANDE)	0.000000	18.21429
CANTIDAD(C, CHICO)	0.000000	27.32143
CANTIDAD(C, MEDIANO)	148.6607	0.000000
CANTIDAD(C, GRANDE)	84.37500	0.000000